



Software Mathematica na střední škole




Jakub Šerých, serych@panska.cz

Využití ve výuce “on the fly”

Občas se ve výuce narazí na nějakou okamžitou otázku, kterou je třeba studentům objasnit.

Například aproximace funkce sinus lineární funkcí v okolí nuly:

```
Plot[{Sin[x], x}, {x, -1, 1}, AspectRatio -> Automatic]
```



... nebo s použitím jednoduché interaktivity


```
Manipulate[ $\frac{a - \text{Sin}[a]}{a} 100$ , {a, -1, 1}]
```



Jiným příkladem může být vizualizace řešení jednoduché nerovnice s absolutní hodnotou:


Graficky najděte řešení $|x-3|-5 \leq 0$

```
Plot[Abs[x - 3] - 5, {x, -10, 10}]
```



... a s mírným rozšířením o interaktivní ovládání:

```
Manipulate[
  Plot[Abs[Abs[x - vodor] - svis2] - svis, {x, -12, 12},
    AspectRatio -> Automatic, AxesOrigin -> {0, 0}],
  {vodor, -8, 8},
  {svis, -8, 8},
  {svis2, -5, 5}
]
```



Složitější ukázky je ale lepší předem pečlivěji propracovat

```

parametry = {
  {{{-5, 5}, {0, 6}}, {{-5, 5}, {-1, 1}}, None}, (* konstanta*)
  {{{-5, 5}, {-5, 5}}, {{-5, 5}, {0, 2}}, None}, (* x *)
  {{{-10, 10}, {0, 100}}, {{-10, 10}, {-20, 20}}, None}, (* x^2 *)
  {{{-10, 10}, {-1000, 1000}}, {{-10, 10}, {0, 300}}, None}, (* x^3 *)
  {{{-10, 10}, {-10, 10}}, {{-10, 10}, {-100, 0}}, x == 0}, (* 1/x *)
  {{{-2, 4}, {0, 260}}, {{-2, 4}, {0, 300}}, None}, (* 4^x *)
  {{{0.0001, 100}, {-5, 5}}, {{0.0001, 100}, {0, 20}}, None}, (* log x *)
  {{{0, 4 Pi}, {-1, 1}}, {{0, 4 Pi}, {-1, 1}}, None}, (* sin x *)
  {{{0, 4 Pi}, {-1, 1}}, {{0, 4 Pi}, {-1, 1}}, None}, (* cos x *)
  {{{0, 2 Pi}, {-50, 50}}, {{0, 2 Pi}, {0, 150}}, Cos[x] == 0}, (* tangens *)
  {{{0.001, 2 Pi}, {-50, 50}},
   {{0.001, 2 Pi}, {-150, 0}}, Sin[x] == 0} (* cotangens *)
};

Manipulate[
  par = Switch[fs,
    fcon, parametry[[1]],
    fx, parametry[[2]],
    fx2, parametry[[3]],
    fx3, parametry[[4]],
    fillomx, parametry[[5]],
    f4nax, parametry[[6]],
    flog, parametry[[7]],
    fsin, parametry[[8]],
    fcos, parametry[[9]],
    ftan, parametry[[10]],
    fcot, parametry[[11]]
  ];
  fce[x_] = Switch[fs,
    fcon, 5,
    fx, x,
    fx2, x^2,
    fx3, x^3,
    fillomx, x^-1,
    f4nax, 4^x,
    flog, Log[x],
    fsin, Sin[x],
    fcos, Cos[x],
    ftan, Tan[x],

```


```

    fcot, Cot[x]
  ];
xmin = par[[1, 1, 1]];
xmax = par[[1, 1, 2]];
delX = (xmax - xmin) / 10;
smer = D[fce[x], x] /. x -> bodX;
deriv[x_] = D[fce[x], x];

Grid[{
  {Style["f(x) = " <> ToString[TraditionalForm[fce[x]]], "Subsubtitle"}},
  {Show[
    Plot[fce[x], {x, xmin, xmax},
      AxesOrigin -> {0, 0},
      Axes -> {True, True},
      PlotRange -> par[[1]],
      PlotStyle -> Thick,
      Exclusions -> par[[3]],
      ExclusionsStyle -> Dashing[Small],
      ImageSize -> {380, 280}],
    Graphics[
      {PointSize[Large], Orange, Point[{bodX, fce[bodX]}],
        Line[{bodX - delX, fce[bodX] - smer delX}, {bodX + delX, fce[bodX] + smer delX}]}
    ]}],
  {Style["f'(x) = " <> ToString[TraditionalForm[deriv[x]]], "Subsubtitle"}},
  {Plot[deriv[x], {x, xmin, bodX + 0.001},
    AxesOrigin -> {0, 0},
    Axes -> {True, True},
    PlotRange -> par[[2]],
    PlotStyle -> Thick,
    Exclusions -> par[[3]],
    ExclusionsStyle -> Red,
    ImageSize -> {380, 280}
  ]}],
  {{fs, ftan, "Výběr funkce:"},
  {fcon -> "konst. fce.",
    fx -> "x",
    fx2 -> "x2",
    fx3 -> "x3",
    flomx -> "x-1",
    f4nax -> "4x",
    flog -> "log x",
    fsin -> "sin x",
    fcos -> "cos x",
    ftan -> "tg x",
    fcot -> "cotg x"}, ControlType -> PopupMenu}],

```

```
{bodX, 1.5, "Derivace v bodě:"},  
Dynamic[xmin], Dynamic[xmax], Appearance -> "Labeled"},  
SaveDefinitions -> True]
```



Z práce studentů

Ukázka použití diferenciálních rovnic


Převzato z ročníkové práce studentů 3. ročníku (V. Myslivec, C. Šup 2008/2009)

`m = 2; g = 9.8; $\alpha = \frac{\pi}{4}$; v0 = 10;`

```
Manipulate[
  SilaX[vx_] := -k vx Norm[vx];
  SilaY[vy_] := -g m - k vy Norm[vy];

  difX =
    NDSolve[{m rX''[t] == SilaX[rX'[t]], rX[0] == 0, rX'[0] == v0 Cos[ $\alpha$ ]}, rX, {t, 0, 10}];
  difY = NDSolve[{m rY''[t] == SilaY[rY'[t]], rY[0] == 0, rY'[0] == v0 Sin[ $\alpha$ ]},
    rY, {t, 0, 10}];

  Show[
    Plot[
      X Tan[ $\alpha$ ] -  $\frac{9.8}{2} (X / (v0 Cos[\alpha]))^2$ ,
      {X, 0, 11},
      PlotRange -> {{0, 12}, {0, 4}},
      AspectRatio ->  $\frac{1}{3}$ ,
      PlotStyle -> {Darker[Blue], Thickness[0.01]},
      ImageSize -> {900, 300}
    ],
    ParametricPlot[
      {rX[t] /. difX[[1]], rY[t] /. difY[[1]]},
      {t, 0, 10},
      PlotRange -> {{0, 12}, {0, 4}},
      AspectRatio ->  $\frac{1}{3}$ ,
      PlotStyle -> {Orange, Thickness[0.005]},
      ImageSize -> {900, 300}
    ]
  ],
  {{k, 0.1, "Odporová konstanta"}, 0, 0.5, 0.01, Appearance -> "Labeled"},
  SaveDefinitions -> True
]
```




Jedna věta z jejich obhajoby této práce: “Zde je nutné zmínit, že na začátku 3. ročníku jsme skoro ani nevěděli jak se derivuje, ale díky Wolfram Mathematice jsme byli schopni použít diferenciální rovnice.”

Využití v odborných předmětech

Vykreslení elektrického pole


Přímo v nápovědě programu *Mathematica* najdeme tento jednoduchý, ale mocný příklad:

```
Manipulate[
  ContourPlot[q1 / Norm[{x, y} - p[[1]]] + q2 / Norm[{x, y} - p[[2]]] ,
    {x, -2, 2}, {y, -2, 2}, Contours -> 10],
  {{q1, -1}, -3, 3},
  {{q2, 1}, -3, 3},
  {{p, {{-1, 0}, {1, 0}}}, {-1, -1}, {1, 1}, Locator},
  Deployed -> True]
```



Můžeme ho celkem bez velkého přemýšlení ihned rozšířit

```
Manipulate[
  ContourPlot[
    q1 / Norm[{x, y} - p[[1]]] + q2 / Norm[{x, y} - p[[2]]] + q3 / Norm[{x, y} - p[[3]]],
    {x, -2, 2}, {y, -2, 2}, Contours -> 20],
  {{q1, 1}, -3, 3},
  {{q2, 1}, -3, 3},
  {{q3, -1}, -3, 3},
  {{p, {{-1, 0}, {1, 0}, {0, -1}}}, {-1, -1}, {1, 1}, Locator},
  Deployed -> True]
```



Pokud si s ním někdo trochu více pohraje, může to dopadnout například takto:

<http://demonstrations.wolfram.com/ElectricFieldsForThreePointCharges/>

Amplitudová modulace

<http://demonstrations.wolfram.com/AmplitudeModulation/>

Typy magnetů v urychlovači LHC:

<http://demonstrations.wolfram.com/MagnetTypesInParticleAccelerators/>

Možnosti využití Mathematicy pro různé studentské projekty

Wolfram Data Drop

Datové úložiště se snadným přístupem z programu *Mathematica* a širokými možnostmi vkládání dat

Vytvoření Databinu

```
(* CreateDatabin["Interpretation" -> {"id" -> "Number", "loc" -> "Location",  
  "distance" -> "Number", "bearing" -> "Number", "deviation" -> "Number",  
  "stations" -> "Number", "part" -> "Number", "time" -> "DateTime"}]
```

*)

```
Databin[ Short ID: 6hoz1_m2  
Entry count: 9830 ]
```

Otevření Databinu a načtení dat

```

In[3]= mujbin = Databin["6hoz1_m2"]
valbin = Values[mujbin];
(* transformace dat do potrebného formátu *)
velka = Normal[valbin]
tabulka = Transpose[{
  "time" /. velka[[2]],
  "id" /. velka[[6]],
  "loc" /. velka[[3]],
  "distance" /. velka[[7]],
  "deviation" /. velka[[5]],
  "bearing" /. velka[[4]],
  "stations" /. velka[[1]],
  "part" /. velka[[8]]
}];
Dimensions[tabulka]

(* utřídění dat *)
(* Blesky které detekovala i moje stanice *)
moje = Cases[tabulka, {_, p_} /; p == 1];
Length[moje]

(* Blesky na kterých se moje stanice nepodílela *)
cizi = Cases[tabulka, {_, p_} /; p == 0];
Length[cizi]

(* moje blesky seříděné podle vzdálenosti *)
mojeNej = Sort[moje, #1[[4]] > #2[[4]] &];

(* moje blesky seřazené podle azimutu *)
mojePol = Sort[moje, #1[[6]] > #2[[6]] &];

(* malý vzorek nejvzdálenějších a nejbližších blesků *)
vzorek = Join[mojeNej[[1 ;; 8]], mojeNej[[-8 ;; -1]]];
Grid[Prepend[vzorek, Style[#, {Red, Bold, 14}] & /@
  {"time", "id", "loc", "distance", "dev", "bear", "sta", "p"}], Frame -> All]

```

Out[3]= Databin [ Name: Blesky
Entry count: 9830]

```
{stations → {160, 103, 87, 64, 80, 174, 41, 163, 42, 66, 100, 137, 194, 33, 252, 83, 99,  
126, 70, 147, 189, 353, 33, 40, 135, 35, 77, 225, 307, 97, 112, 73, 226, 336, 173,  
332, 118, 59, 58, 51, 164, 274, 119, 170, 268, 57, 126, 32, 191, 38, 131, 33,  
Out[5]= ... 17178 ... , 142, 11, 42, 14, 16, 36, 35, 16, 76, 39, 117, 19, 10, 120, 54, 236, 17,  
26, 13, 11, 100, 70, 11, 143, 50, 15, 78, 167, 157, 17, 29, 27, 15, 25, 30, 61, 28,  
119, 20, 128, 68, 27, 64, 24, 112, 38, 39, 15, 69, 10, 37, 18}, ... 6 ... , ... 1 ... }
```

large output | [show less](#) | [show more](#) | [show all](#) | [set size limit...](#)

```
Out[7]= {17282, 8}
```

```
Out[9]= 17282
```

```
Out[11]= 0
```

time	id	loc	distance	dev	bear	sta	p
 Sun 2 Aug 2015 21:13:27 GMT-5.	13 873 587	GeoPosition[{-21.0257, 43.9766}]	8 419 423	4228	160.9	160	1
 Thu 9 Jul 2015 02:32:07 GMT-5.	10 615 391	GeoPosition[{-25.5111, 14.0371}]	8 398 940	4489	180.2	103	1
 Tue 7 Jul 2015 16:34:24 GMT+12.	9 847 890	GeoPosition[{-20.5156, 43.1186}]	8 336 916	4018	161.4	87	1
 Sat 1 Aug 2015 16:44:01 GMT+12.	13 725 193	GeoPosition[{-24.5759, 21.5671}]	8 324 921	2183	175.2	64	1
 Sat 9 May 2015 00:14:14 GMT+2.	1 498 587	GeoPosition[{-23.8641, 21.1469}]	8 242 596	2884	175.5	80	1
 Wed 8 Jul 2015 20:43:44 GMT+4.	10 522 464	GeoPosition[{-20.3347, 38.329}]	8 170 041	3193	164.2	174	1
 Fri 10 Jul 2015 05:08:51 GMT+2.	10 695 594	GeoPosition[{-21.5355, -2.8302}]	8 136 326	3848	191.5	41	1
 Thu 6 Aug 2015 15:18:34 GMT+2.	14 260 544	GeoPosition[{-18.3456, 43.6596}]	8 126 991	4235	160.8	163	1
 Wed 8 Jul 2015 02:00:16 GMT-5.	10 196 607	GeoPosition[{50.0445, 14.3933}]	2850	1453	333.5	156	1
 Mon 27 Apr 2015 18:58:01 GMT-5.	752 838	GeoPosition[{50.0171, 14.4389}]	2048	3993	104.2	16	1
 Sun 26 Apr 2015 15:43:43 GMT-5.	594 418	GeoPosition[{50.032, 14.3875}]	2046	2571	304.5	196	1
 Tue 23 Jun 2015 18:28:49 GMT-5.	7 764 637	GeoPosition[{50.0115, 14.3886}]	1963	4485	235.1	67	1
 Mon 8 Jun 2015 21:37:43 GMT-5.	4 951 885	GeoPosition[{50.0161, 14.4342}]	1759	1728	110.4	227	1
 Wed 8 Jul 2015 02:02:33 GMT-5.	10 196 991	GeoPosition[{50.0229, 14.4317}]	1478	3923	84.5	40	1
 Wed 8 Jul 2015 02:04:22 GMT-5.	10 197 405	GeoPosition[{50.0232, 14.4287}]	1270	991	81.9	21	1
 Wed 8 Jul 2015 02:02:02 GMT-5.	10 196 899	GeoPosition[{50.0282, 14.4126}]	738	2649	8.	88	1

Out[15]=


A teď už pár ukázek možných výstupů

Všechny nalovené blesky na mapě

In[18]=

```
GeoListPlot[mojeNej[[1 ;; 1500, 3]],  
  PlotMarkers → Style["↓", Red, 20], (*GeoRange→{{5,52},{-33,52}},*)  
  GeoGridLines → Quantity[10, "AngularDegrees"], GeoProjection → "Mercator",  
  ImageSize → Large]
```

Out[18]=




Mapa s vyznačenou vzdáleností od mé stanice

```

In[22]= (* Tabulka souřadnic blesků zachycených mojí stanicí *)
mylatlon = Transpose[{QuantityMagnitude[Latitude[moje[[1 ;; 1500, 3]]]],
  QuantityMagnitude[Longitude[moje[[1 ;; 1500, 3]]]]};
(* Souřadnice mé stanice *)
stanice = GeoPosition[{50.021606, 14.411118`}];
(* vykreslení mapy *)
GeoGraphics[{Dashed, Thick,
  Table[GeoCircle[Prague(city), Quantity[r, "km"]], {r, 1000, 9000, 1000}],
  PointSize[Large], Blue, Point[stanice], PointSize[Large],
  (* Blue, Point[GeoPosition[cilatlon]], *)
  Red, Point[GeoPosition[mylatlon]]},
GeoCenter → GeoPosition[{00., 39.}],
GeoRange → Quantity[6000, "km"],
GeoProjection → "Mercator",
GeoBackground → Automatic,
ImageSize → {600, 600}]

```

Out[24]=




Blízké blesky ve větším detailu

```

In[34]= mylatlon = Transpose[{QuantityMagnitude[Latitude[moje[[All, 3]]]],
    QuantityMagnitude[Longitude[moje[[All, 3]]]]}];
GeoGraphics[{Dashed, Thick,
    Table[GeoCircle[stanice, Quantity[r, "km"]], {r, 10, 50, 10} (* {r, 20, 400, 20} *)],
    PointSize[Large], Blue, Point[stanice], PointSize[Medium],
    (* Blue, Point[GeoPosition[cilatlon]], *)
    Red, Point[GeoPosition[mylatlon]]},
    GeoCenter → GeoPosition[stanice],
    GeoRange → Quantity[40, "km"],
    GeoProjection → "Mercator",
    GeoBackground → Automatic,
    ImageSize → {600, 600}]

```



Out[35]=

Díky za pozornost!